

# Word for Windows Reference Database

Version 1.0

Paul Butcher  
Department of Computer Science  
University of York  
Heslington  
YORK, YO1 5DD

paulb@uk.ac.york.minster

This is a slightly updated release of my reference database macro package for Word for Windows. It contains a couple of bug fixes over version 0.1, and has been upgraded to work with Word for Windows 2.0. Note, this probably means it won't work with Word for Windows 1.0 any more (I haven't checked). This version gets the coveted "1.0" because I'm actually fairly convinced that it works — 0.1 on the other hand was decidedly flaky.

This document contains two Word for Windows macros which implement a database. It is intended to perform the same sort of job as either BibT<sub>E</sub>X or UN\*X refer, however as I wrote it primarily for my own use it doesn't try to duplicate all of their functionality. In particular, this package doesn't "understand" references in the way that either of the above do, and doesn't provide any formatting of them. Nor are there reference "styles" (e.g. \bibliographystyle equivalents), so you can't produce an abbreviated bibliography from the same source as a full bibliography. Another fairly major disadvantage is speed – Word insists on updating the screen while running a macro, and this causes things to go rather slowly. If anyone knows a way round this, I'd be very interested to hear it!

## Installation

The text of the macros is shown below. Two bookmarks are set up to point to the two macros (use Edit / Go To). Simply cut and paste them into an appropriate template. If you want to, they can be bound to a menu entry or a key. The glossary entry "RefFile" should be set up to point to the reference database you want to use (this is simply a W4W document). If you want references in footnotes to be resolved, then define the glossary entry "FootnoteRef"

## Constructing a Reference Database

Reference databases are simply a set of paragraphs of the form

[ ]      Bloggs, G. *A Very Interesting Paper*, Journal of Interesting Things, 2(12):10–32, Dec 1991.

The only bit of this that is required is the sequence of fields at the start (use View / Field Codes to see them). The first field sets a bookmark "Bloggs91a" to be the next number in the sequence "Reference". The next field references it, so that the number appears in the bibliography itself. The bookmark name may take any form, however it is sensible to choose a mnemonic one. The macro "InsertBibEntry" will create entries of this form. The text after these fields may take any form whatsoever. The "ResolveRefs" macro simply inserts this text verbatim into the document.

## Resolving References

When writing a document which contains references, simply place a field of the form [ ] (use View / Field Codes again) in the document. Place a bookmark "References" in the document where you want the bibliography to appear. Then run the "ResolveRefs" macro, and watch it all happen.

## Guarantee

None whatsoever! If these macros trash your hard disk, tough! I'll try to fix any bugs you might find, but I'm supposed to be writing a thesis, so I may not have much time.

## Distribution

Feel free to distribute this document freely. All I ask is that you keep my name in the comments of the macros and at the top of this document. If you make any improvements, I'd love to see them.

## Miscellany

- The "References" bookmark no longer needs to contain any text. Version 0.1 would also sometimes delete this bookmark at the start of the "ResolveRefs" macro (I never did work out why it was doing it, but I now use a completely different method which seems not to have this problem).
- Version 0.1 would sometimes duplicate references which were referred to more than once. This was due to me not adequately understanding how booleans work in WordBasic. This version fixes that problem.
- The bookmarks "RefEnd" and "RefTmp" are used by the "ResoveRefs" macro, so it's probably not a good idea to use them in your document. Don't delete the RefEnd bookmark either (as it's name suggests it marks the end of the references inserted by a previous execution) or you'll end up with your old bibliography as well as the new.

```
' ResolveRefs
' =====
'
' © 1991 Paul Butcher.
'
' Description: Resolves any references within a document by reading them from a reference database.
' Dependencies:
'     The glossary entry "RefFile" must be set to point to the reference database.
'     References must be fields of the form {ref <name>}.
'     Reference database entries must be of the form;
'         [{set <name> {seq Reference}}{ref <name>}] . . .
'     The bookmark "References" must be defined in the document .
'     If the glossary entry "FootnoteRef" exists, then references in footnotes will be resolved.
```

```
Dim Shared Refs$(256) ' Array where refs are stored.
```

```
Sub MAIN
' Delete the old bibliography
' =====
logOldFootnote = ViewFootnotes()
ViewFootnotes 0
If Not ExistingBookmark("References") Then
    MsgBox "No References Bookmark", "Error", 48
    ViewFootnotes logOldFootnotes
    Goto DoneResolveRefs
End If
If ExistingBookmark("RefEnd") Then
    EditGoTo .Destination = "References"
    ExtendSelection
    EditGoTo .Destination = "RefEnd"
    Cancel
    EditClear
    InsertBookmark "RefEnd", .Delete
End If
```

```

' Make a list of references in document
' =====
StartOfDocument
ShowAll 1
ViewFieldCodes 1
x = 0
f = 1
While x < 256
    If NextField() = 0 Then
        If f = 1 And ExistingBookmark("FootnoteRef") Then
            ViewFootnotes 1
            f = 0
            StartOfDocument
            Goto NotRef
        Else
            Goto DoneRefs
        End If
    End If
    t$ = Mid$(Selection$, 2, Len(Selection$) - 2)
    If LCase$(Left$(t$, 3)) <> "ref" Then Goto NotRef
    t$ = Right$(t$, Len(t$) - 4)
    While Left$(t$, 1) = " "
        t$ = Right$(t$, Len(t$) - 1)
    Wend
    If Not IsIn(t$, x) Then
        x = x + 1
        Refs$(x) = t$
    End If
NotRef:
Wend
DoneRefs:                                     ' x contains no. of references found
winDoc = Window()

' Open the reference database, and place the selected references on the spike.
' =====
refFile$ = GetGlossary$("RefFile")
FileOpen .Name = refFile$
SetGlossary "Spike", ""
For y = 1 To x
    If ExistingBookmark(Refs$(y)) Then
        EditGoTo .Destination = Refs$(y)
        ParaUp 1
        ParaDown 1, 1
        EditCopy
        Spike
        EditPaste
    End If
Next y
FileClose 2
' Return to the document, and insert the references
' =====
Select Case winDoc
Case 1
    Window1
Case 2
    Window2
Case 3
    Window3
Case 4

```

```

        Window4
    Case 5
        Window 5
    Case 6
        Window 6
    Case 7
        Window 7
    Case 8
        Window 8
    Case 9
        Window 9
    End Select
    ShowAll 0
    ViewFieldCodes 0
    EditGoTo .Destination = "References"
    UnSpike
    ExtendSelection
    EditGoTo .Destination = "References"
    Cancel
    EditReplace .Find = "^p^p", .Replace = "^p", .WholeWord = 0, .MatchCase = 0, \
.ReplaceAll = 1, .Format = 0
    EditReplace .Find = "^p^p", .Replace = "^p", .WholeWord = 0, .MatchCase = 0, \
.ReplaceAll = 1, .Format = 0
    ToolsSorting .Separator = 1, .FieldNum = 2
    UpdateFields
    InsertBookmark "RefTmp"
    SetStartOfBookmark "RefTmp", "References"
    SetEndOfBookmark "RefTmp", "RefEnd"
    InsertBookmark "RefTmp", .Delete
    StartOfLine
    StartOfDocument 1
    UpdateFields
    If ExistingBookmark("FootnoteRef") Then
        OtherPane
        EditSelectAll
        UpdateFields
    End If
    ViewFootnotes logOldFootnote
DoneResolveRefs:
End Sub

Function IsIn(r$, n)                                ' Returns 1 if r$ is in first n references.
    For x = 1 To n
        If Refs$(x) = r$ Then
            IsIn = - 1
            Goto DoneIsIn
        End If
    Next x
    IsIn = 0
DoneIsIn:
End Function

' InsertBibEntry
' =====
'
' © 1991 Paul Butcher.
'
' Description: Places a formatted reference into a document.

Sub MAIN

```

```
tag$ = InputBox$("Tag: ", "Reference tag")
Insert "["
InsertFieldChars
Insert "set " + tag$ + " "
InsertFieldChars
Insert "seq Reference"
CharRight 2
InsertFieldChars
Insert "ref " + tag$
CharRight 1
Insert "]"
InsertPara
CharLeft 1
StartOfLine 1
UpdateFields
EndOfLine
End Sub
```